ANALYSIS OF SOIL MOISTURE EXTRACTION ALGORTIHM
USING DATA FROM AIRCRAFT EXPERIMENTS

ERT Document No. P-A826

Final Report

June 1981

Prepared for

National Aeronautics & Space Administration
Goddard Space Flight Center
Greenbelt, Maryland 20771

Contract No. NAS5-26361

Prepared by

Hsiao-hua K. Burke and Jean-Hsien Ho

Environmental Research & Technology, Inc.
696 Virginia Road
Concord, Massachusetts 01742

ABSTRACT

In this study, a soil moisture extraction algorithm was developed using a statistical parameter inversion method. Data sets from two aircraft experiments were utilized for the test. Both data sets contain multifrequency microwave radiometric data as well as surface temperature and soil moisture information.

Results show that by using L, C and X band radiometer data, the surface and near surface ($\leq$5 cm) soil moisture content can be extracted with accuracy of approximately 5-6% for bare fields and fields with grass cover. It is demonstrated that this technique is of potential use for handling large amounts of remote sensing data from space.

TABLE OF CONTENTS

1. INTRODUCTION

The concept of soil moisture monitoring utilizing passive microwave sensors has been investigated both theoretically and experimentally. Although both the dielectric properties and radiative transfer processes are relatively well understood, the problem still remains complex as effects due to various roughness scales, vegetation cover, and other mixed terrain features have not been fully modeled and implemented. The objective of this study is to take a different approach; instead of investigating radiometric responses, feasibility of using a statistical inversion scheme for soil moisture retrieval from existing data sets is explored.

The basic tool adopted for retrieving soil moisture information from radiometric data is the Statistical Parameter Inversion Method. In essence, the method seeks out significant statistical correlations between the measured data and their associated parameters. It is a general regression technique which minimizes, in the statistical sense, the mean square error between the estimated and observed values of the parameters of interest. The model is described in Section 2 and the computer codes are included in Appendix A.

Data sets from two aircraft experiments were utilized. Both experiments were carried out with multi-frequency microwave sensors together with detail ground truth measurements. One set of data was collected over agriculture fields near Phoenix, Arizona during March 1975. The statistical retrieval was carried out for data from bare fields. Radiometric data of 21 cm, 2.8 cm and 1.67 cm, as well as surface temperature information, are used to retrieve soil moisture contents in the 0-2 cm and 2-5 cm layers. Another set of data was similarly collected over two semi-arid areas - Chickasha, Oklahoma and Riesel, Texas - during May 1978. The fields were mostly grass-covered. Radiometric data of 21 cm and 5 cm as well as surface temperature measurements are used to retrieve volumetric soil moisture contents in the 0-2.5 cm and 2.5-5 cm layers. Section 3 summarizes the data sets and Appendix B illustrates the radiometric responses as a function of soil moisture content for the Chickasha/Riesel experiment. Further detail of the data can be found in separate reports by Burke (1980) and Jackson et al. (1980).

The results are summarized in Section 4. Over bare fields, the soil moisture content can be retrieved with an rms error of approximately 5% using L and X band measurements as demonstrated from data of the Phoenix experiment. Over fields with grass cover, the rms error for similar retrieval is approximately 6% using L and C band measurements as shown from data of the Chickasha/Riesel experiment.

In conclusion, the statistical retrieval technique can be valuable for handling large amount of data provided the correlation between data and parameter is well established either from well controlled experiments or theoretical simulation. The technique was demonstrated with two aircraft experiment measurements. There are also large amounts of multifrequency satellite data available. The next step should be the utilization of this type of data for feasibility of soil moisture retrieval over large areas.

2. DESCRIPTION OF THEORY AND MODEL

In this study, the basic tool adopted for retrieving soil moisture information from radiometric data is the Statistical Parameter Inversion Method. This technique was originally developed by Westwater and Strand (1965) and modified by Waters and Staelin (1968) and Gaut (1967). It has been used extensively in many remote sensing studies and applications. Recently, it was employed for development of operational retrievals of geophysical parameters ranging from ocean surface wind speed to cloud and rain conditions for a seven-channel microwave radiometer (Special Sensor Microwave/Imager (SSM/I)). The SSM/I system will be flown on a future mission of the Defense Meteorological Satellite Program.

In essence, the method seeks out significant statistical correlations between the measured data and their associated parameters. It is a general regression technique which minimizes, in the statistical sense, the mean square error between the estimated and observed values of the parameters of interest. To do this, it starts from an ensemble of simulated cases. These cases are represented by radiometric data associated with the physical parameters. The resulting correlations are contained in the inversion matrix, which is a set of coefficients.

The underlying assumption for this technique is that there exist some linear combination of data elements which will provide a useful estimate of the desired parameters. That is to say:

$$p_i^* = \sum_{j=1}^{n} D_{ij} d_j$$

where p_i^* is the estimate of the parameter p_i, d_j's are the brightness temperatures and D_{ij} is the inversion matrix. This can be written vectorially as:

$$\underline{p} = \underline{\underline{D}} \cdot \underline{d}$$

The determining conditions on the inversion matrices results from minimizing the square of the deviation of the estimated from the actual values of the parameters over the entire ensemble of simulated scenes. It can be shown as:

$$\underline{\underline{D}} = \underline{\underline{C}}(\underline{p},\underline{d}) \cdot \underline{\underline{C}}^{-1} (\underline{d},\underline{d})$$

where $\underline{\underline{C}}(\underline{p},\underline{d})$ and $\underline{\underline{C}}(\underline{d},\underline{d})$ are the correlation matrices defined as

$$C_{ij}\ (\underline{x},\underline{y}) = \langle x_i, y_i \rangle$$

As can be seen, the Statistical Parameter Inversion Method relies on a priori information defining the correlation between the predicted data vector ($\underline{d}$) and the predicted geophysical parameter vector ($\underline{p}$).

Another feature can be used in the inversion process to "tune" the inversion matrix to the average value of each parameter. To do this, the data vector is transformed like:

$$\underline{\psi}\ (\underline{d}) = (1,\ d_1 - \overline{d}_1,\ d_2 - \overline{d}_2,\ \ldots,\ d_n - \overline{d}_n)$$

The effect of this transformation is to add a column to the inversion matrix which contains the ensemble averages of the parameters. Though the resulting inversion is still linear in the T_B's, $\underline{\underline{D}}$ is now focused on correlated deviations from the expected "mean" values of the parameters. For example, in retrieving soil moisture, the expected mean value can be that usually occurring or expected for the season. However, data for this study were acquired from a number of experiments which included a broad range of soil moisture conditions. Therefore, the "tuning" process to a mean or most probable case was not used.

Problems arise which can degrade the inversion results when there are non-linearities in the $\underline{d}$ to $\underline{p}$ relationships. Various modifications can be developed to correct for this problem (e.g. Staelin, 1980; Wilheit and Chang, 1980). In the remote sensing of soil moisture in the passive microwave region, the brightness temperature to soil moisture relationship is approximately linear; therefore, treatments for non-linearities were not considered for this study either.

Appendix A lists the FORTRAN codes of the computer program developed for this study. It consists of five subroutines with their main functions listed as follows:

1) MAIN-Program SPIM: the Statistical Parameter Inversion Method is used to obtain the inversion matrix, upon option, results are then evaluated by computing the standard deviation between estimated (computed) and actual parameter values;

2) INDATA: to set up data and parameters for inversion matrix calculation;

3) CORMET: to calculate the correlation matrices from the known parameter and data sets;

4) MINV: an adopted IBM application program to perform matrix inversion using the standard Gauss-Jordan method; and

5) GMPRD: to multiply two matrices to form a resultant general matrix.

3. DESCRIPTION OF DATA

In this study, data from two aircraft experiments were utilized. One was carried out for agriculture fields near Phoenix, Arizona during March 1975. Another set of data was collected in two semi-arid areas - Chicasha, Oklahoma and Riesel, Texas during May 1978. Data from the Phoenix experiment was described and analyzed in a separate report by Burke (1980). Data from the Chickasha and Riesel experiment was described in Jackson et al. (1980). The following is a summary of the two data sets.

3.1 Data from the Phoenix Experiment

During March 1975, an aircraft mission consisting of four flights over the Phoenix, Arizona test site was conducted for the purpose of studying the use of microwave radiometers for the remote sensing of soil moisture . The investigators involved in this mission came from NASA, the Agricultural Research Service of USDA, the University of Arkansas, and Texas A&M University. This mission consisted of predawn and midday flights on 18 and 22 March 1975. There were radiometers operating at wavelengths of 0.8, 1.3, 1.67, 2.8 and 21 cm. The 2.8 cm radiometer is a dual-polarized conically scanning radiometer operating at a fixed look angle of 50°. The other radiometers were non-scanning but could have their nadir look angles varied. In addition to the microwave instruments the scientific package included a thermal infrared radiometer for measuring surface temperature. Three passes were taken over each field; one at a nadir angle of $\Theta = 0°$ and two at a nadir angle of $\Theta = 40°$ alternating the polarization sensitivity of the 21 cm antenna.

Ground measurements were made in 46 fields. Twenty-eight were without vegetative cover and 18 had vegetative covers of either alfalfa or wheat. The fields, which have an area of 16 hectares (400 x 400 m), were arranged in pairs to provide a uniform target 800 meters wide. The soil moisture sampling procedures for this mission included measurements of the moisture content and temperature in each of the following soil layers: 0-1 cm, 1-2 cm, 2-5 cm, 5-9 cm, and 9-15 cm. The results included data from a variety of moisture conditions due to the irrigation and drying cycles of the fields.

The data analysis was carried out for 21 cm, 2.8 cm and 1.67 cm measurements. Analysis of data from 1.3 cm and 0.8 cm were not carried out as they were relatively insensitive to soil moisture content. Responses to both bare and vegetated fields for nadir and polarized conditions were investigated in detail in Burke (1980).

In this study, soil moisture contents in the 0-2 cm and 2-5 cm layers are taken to be the two parameters to be retrieved from statistical inversion; they represent the surface and near surface moisture conditions. Only bare fields are treated. The average physical temperature of the surface is used to normalize the brightness temperature such that morning and afternoon measurements can be utilized simultaneously.

3.2 Data from the Chicasha and Riesel Experiment

Cooperative investigations were conducted during May 1978 by the National Aeronautics and Space Administration (NASA) and the U.S. Department of Agriculture (USDA) as part of a project to evaluate remote sensing in hydrologic studies with primary emphasis on measurements. Ground observations and aircraft remote sensing experiments were conducted in two semi-arid areas - Chicasha, Oklahoma and Riesel, Texas. Participants in the study were the NASA Goddard Space Flight Center and the USDA-SEA-AR Hydrology Laboratory, Southern Plains Watershed and Water Quality Laboratory and the Grassland, Soil, and Water Laboratory.

Three successful flights were made on May 1, 12 and 30, over the two test sites. The study sites and watersheds in Oklahoma were in the Washita River Experimental Watershed, Chickasha, area. Most watersheds had a dense grass cover; some were bare or wheat fields. The study site located in the central part of Texas near Riesel was also on an experimental watershed area. The land cover varied from almost bare soil to a very dense vegetative cover. Some fields experienced considerable vegetation growth between the first flight and the last flight.

Sensor data included those from photography, thermal infrared radiometer, passive microwave sensor system of L-band (21 cm), C-band (5 cm), X-band (2.8 cm), Ku-band (1.67 cm), K-band (1.35 cm) and Ka-band (.81 cm) with look angles at 0° and/or 40°, and scatterometer operating at L, C and P bands. For fields from both study sites, all soil moisture samples were weighed, oven-dried and weighed again to determine their gravimetric soil moisture. Bulk density samples from each field were then used to determine the volumetric soil moisture. Soil moisture information in the 0-2.5 cm, 2.5-5 cm, 5-10 cm and 10-15 cm was obtained (Jackson et al., 1980).

In this study, volumetric soil moisture contents in the 0-2.5 cm and 2.5-5 cm are taken as the two parameters to be retrieved. Surface temperature obtained from the infrared sensor is used to normalize different conditions from various days and sites. Radiometric data from L, C, X and Ku bands are utilized. Since there was no previous study on analysis of the data, brightness temperatures at these wavelengths are plotted as a function of the 0-2.5 cm layer volumetric moisture content (Appendix A) for a general feeling of the radiometric response.

4. RESULTS AND CONCLUSIONS

Figure 4-1 demonstrates an ideal approach of parameter (e.g. soil moisture) retrieval from remote sensing data. An ensemble of the physical system is set up first. This system contains a broad and representative range of model cases, each with a different combination of soil moisture condition, surface roughness, vegetation coverage, temperature, atmospheric contamination and other factors that can affect the radiometric signature. Applying the radiative transfer and other model computations, an ensemble of simulated radiometric data set is obtained. Correlation matrices of soil moisture and radiometric data as well as the inversion matrix are calculated as outlined in Section 2. Measured radiometric data set is then applied to the inversion matrix to obtain the estimated parameter values. The estimated values can then be compared with the ground truth data for error analysis.

In the present study, the approach is somewhat different due to the following:

1) Data from well controlled experiments are utilized. The data amount is relatively small; variability in soil moisture is broad, but much less significant in other conditions within the same experiment; for example, soil types and physical temperatures. It is thus not applicable to set up a statistical ensemble with model computation of radiometric prediction for the inversion matrix.

2) The main purpose of the study is to test the retrievability of soil moisture condition from radiometric data rather than to test the accuracies of theoretical modeling. Theoretical studies of soil moisture response have been carried out in great extent. In Burke (1980) report, data from the Phoenix experiment showed good agreement with model computation.

As a result, measured data are used both for the derivation of the inversion matrix and also error analysis. The rms error defined by $[\Sigma(p-p^*)^2/N]^{1/2}$ is used where p and p* are the parameter and the estimated value and N is the number of cases.



Figure 4-1 Flow diagram for parameter retrieval using Statistical Parameter Inversion Method

4.1 Results of Soil Moisture Retrieval Utilizing Data from the Phoenix Experiment

As described in Section 3.1, soil moisture contents in the 0-2 cm and 2-5 cm layers are taken as the two parameters to be retrieved. Tests are carried out using (1) five channels of data from 21 cm (0°, vertical and horizontal) and 2.8 cm (vertical and horizontal) and (2) eight channels with additional data from 1.67 cm (0°, vertical and horizontal). Since emissions at 2.8 cm and 1.67 cm from vegetated fields are relatively insensitive to soil moisture contents, retrievals are carried out for bare fields only.

Question arises as to whether the data sets for deriving the inversion matrix and for accuracy test can be the same or they should be independent of each other. The morning measurements from two separate days are used for demonstration. Table 1 shows the difference between using data separately and combined. Data from five channels are utilized in this case. As can be seen, using the same data sets for derivation and test causes the rms error of retrieval to reduce by 1 - 1½% in soil moisture content. Bearing this in mind, the rest of the retrievals are carried out using identical sets of data for both derivation and accuracy test so that maximum number of cases and range of data are used.

TABLE 1

TEST OF RMS INVERSION ERRORS OF SOIL MOISTURE CONTENT FOR VARIOUS WAYS OF USING DATA SETS FOR SET UP AND EVALUATION

Data for Inversion Matrix Set up	Data for Accuracy Test	RMS Inversion Error (% Soil Moisture) 0-2 cm	2-5 cm
Mar 18 AM (14)*	Mar 22 AM (13)	5.6	5.1
Mar 22 AM (13)	Mar 18 AM (14)		
Mar 18 AM Mar 22 AM (27)	Mar 18 AM Mar 22 AM (27)	4.3	4.0

*number in parenthesis refers to the number of cases

Table 2 demonstrates the utilization of both morning and afternoon measurements with five and seven channels of data. Two approaches are tested. In the first case, the brightness temperatures are directly used. In another case, the brightness temperatures are first normalized to the average surface temperatures; 278°K, 300°K, 280°K and 296°K are used for March 18 AM, PM and March 22 AM, PM measurements, respectively. Fifty-five cases are available. For five channels, the rms errors for 0-2 cm and 2-5 cm moisture contents are 5.1% and 4.6%, respectively; a degradation as compared to the 4.3% and 4.0% for morning measurements alone due to the differences in moisture distribution in soil between mornings and afternoons. The improvements from normalizing the brightness temperatures to surface temperatures (T_s) are minimal, from 5.1% and 4.6% to 5.0% and 4.3%.

TABLE 2

RETRIEVAL ERROR OF MOISTURE CONTENT BY WEIGHT IN %
FOR THE PHOENIX EXPERIMENT, 1975

Radiometric Data*	Data Set**	RMS Inversion Error (%) 0-2 cm T_B	0-2 cm T_B/T_S	2-5 cm T_B	2-5 cm T_B/T_S
21, 2.8 cm (5)	Mar 18, 22 AM, PM (55)	5.1	5.0	4.6	4.3
21, 2.8, 1.67 cm (8)	Mar 18, 22 AM, PM (55)	4.7	4.6	4.4	4.2

*number in parenthesis refers to the number of channels
**number in parenthesis refers to the number of cases

Results also show that by utilizing the additional 1.67 cm data, the 0-2 cm error is reduced by about ½% and had little impact on the 2-5 cm results due to its sensitivity to surface moisture content only. The 2-5 cm results have been consistently somewhat better than those of the 0-2 cm. This is due to the less variability in the subsurface moisture conditions between different fields and measurements. Although radiometric data are not more sensitive to subsurface moisture conditions, the statistical inversion error is nevertheless smaller.

4.2 Results of Soil Moisture Retrieval Utilizing Data from the Chickasha/Riesel Experiment

Similar procedure is adopted for the Chickasha/Riesel data set. Volumetric soil moisture contents in the 0-2.5 cm and 2.5-5 cm layers are the two parameters to be retrieved. From the data presented in

Appendix B, it is noted that at 1.67 and 2.8 cm the amount of missing data is high and the sensitivity of the available data to surface moisture is also relatively poor. Data from the five channels of 21 cm (0° and horizontal) and 6 cm (0°, vertical and horizontal) (θ = 40°) are thus utilized. There are cases with missing data in some of the five channels. Since both the derivation of the inversion matrix and also the parameter estimation do not allow missing data, adjustments have to be made. It is decided that if data from more than two out of the five channels are missing, the case is discarded; otherwise, the missing brightness temperature is replaced by the average measured value of that area (Chickasha or Riesel) on that day. As a result, 41 cases are available. As shown in Table 3, the rms inversion errors of the 0-2.5 cm, 2.5-5 cm layers are 7.5% and 6.3%, respectively.

TABLE 3

RETRIEVAL ERROR OF VOLUMETRIC MOISTURE CONTENT IN %
FOR THE CHICKASHA/RIESEL EXPERIMENT, 1978

Radiometric Data*	Data Set**	RMS Inversion Error (%)	
		0-2.5 cm	2.5-5 cm
21, 6 cm (5) T_B's	May 1, 12, 30 (41)	7.5	6.3
21, 6 cm (5) T_B's	May 1, 12, 30 (39)	6.1	5.4
21, 6 cm (5) (T_B/T_S)'s	May 1, 12, 30 (39)	6.0	5.5

*number in parenthesis refers to the number of channels
**number in parenthesis refers to the number of cases

A closer look of the plotted data show that the two fields with highest moisture contents (53.7% and 50.9% for RG 83, RG 88, respectively on May 30) are totally uncorrelated to the rest of the data set. The reason is not known. These two cases are removed and the new retrieval showed rms inversion errors of 6.1% and 5.4% for moisture contents in the 0-2.5 cm and 2.5-5 cm layers, respectively (Table 3).

The same 39 cases are tested with the brightness temperatures normalized to their surface temperature (T_s) measurements. The results are also shown in Table 3 with inversion errors of 6.0% and 5.5%, virtually the same as the previous case.

4.3 Further Discussion

In Sections 4.1 and 4.2, the results of retrieval errors from two aircraft experiments were presented. One major concern is that by using the same data set for both deriving the algorithm and testing it, the uncertainty levels are just those associated with the noise in the data set and do not indicate the uncertainty introduced by the extraction algorithm. Therefore, one more set of tests using the Phoenix data was carried out in order to delineate the difference between uncertainties from the extraction algorithm and the noise in the data. Furthermore, the relative rms error, defined as $[\Sigma(p-p^*/p)^2/N]^{\frac{1}{2}}$, was also computed. The results are shown in Table 4 using normalized brightness temperatures from all eight channels.

TABLE 4

RMS ERROR, BOTH ABSOLUTE AND RELATIVE,
OF SOIL MOISTURE RETRIEVAL, OF THE PHOENIX EXPERIMENT, 1975

Data for Inversion Matrix Setup	Data for Accuracy Test	RMS Inversion Error			
		Absolute (%)		Relative	
		0-2 cm	2-5 cm	0-2 cm	2-5 cm
Mar 18, 22 (55)* AM, PM	Mar 18, 22 (55) AM, PM	4.6	4.2	.41	.27
Mar 18 AM, PM (28)	Mar 22 AM, PM (27)	4.7	5.1	.50	.35

*number in parenthesis refers to the number of cases

As can be seen again, the main source of inversion error is from the noise within the data set rather than the extraction algorithm. It also demonstrated that data sets from the two different days (March 18 and March 22) are compatable with each other. The relative rms errors are generally within 50% (.5) and better for the subsurface retrievals (∿30% or .3). Figure 4-2 is a scatter plot of calculated soil moisture values versus those measured for the second test in Table 4.



Figure 4-2 A scatter plot of calculated soil moisture values versus measured values from the second test case in Table 4

4.4 Conclusions

In this study, the retrievability of soil moisture content from remotely sensed radiometric data is demonstrated. Over bare fields, the moisture content can be retrieved with an rms error of approximately 5% using L (21 cm) and X (2.8 cm) band measurements as shown from data of the Phoenix experiment. Over fields with vegetation cover, the rms error for similar moisture retrieval is approximately 6% using L (21 cm) and C (6 cm) band measurements as shown from data from the Chickasha/Riesel experiment. The results are promising as actual field data are used and only unbiased mathetmatical inversion is applied.

One unexpected result is the lack of improvement when surface temperature measurements are incorporated to normalize the brightness temperatures. This is probably due to the fact that at the wavelengths considered, the physical temperature factor contributed to the signature is an integrated surface-subsurface component, especially at L and C band wavelengths, rather than the surface temperature alone. As all measurements were taken with relatively uniform subsurface temperature, the contribution of the physical temperature may well be within noise level as of other factors (roughness, soil type, etc.).

The handling of missing data from some of the channels is important as it is an inevitable problem of inversion from measured data. In this study, average value of data of similar time and space domains is used. This can be easily adopted for operational use.

Another restriction of the study is the limited amount of useful data. In the Phoenix data set, 55 cases are available; and for the Chickasha/Riesel data set, 39. Ideally, there should be independent data sets for the derivation of inversion and estimation of parameters. An attempt was initiated to correlate the two data sets. It was quickly recognized to be not feasible due to the differences between the two data sets in the wavelengths used, the nature of background (bare versus grassland), and ground truth measurements (layer thicknesses, moisture contents by weight versus volume).

In conclusion, the technique presented and tested in this study proves to be of potential use for large amount of data retrieval. There have been data of this nature from spaceborne sensors; for example, SMMR

(Scanning Multichannel Microwave Radiometer) from both Seasat and Nimbus-7 satellites. After initial analysis of the data, such inversion technique for soil moisture retrievals, as well as for other parameters, should be implemented.

REFERENCES

Burke, H.K., 1980: Evaluation of the Effects of Varying Moisture Contents on Microwave Thermal Emissions from Agriculture Fields, Final Report, Contract No. NAS5-25529, ERT Document No. A245, Environmental Research & Technology, Inc., Concord, Mass.

Gaut, N.E., 1967: Studies of Atmospheric Water Vapor by Means of Passive Microwave Techniques, Ph.D. Thesis, M.I.T., Cambridge, Mass.

Jackson, T.J., T.J. Schmugge, G.C. Coleman, C. Richardson, A. Chang, J. Wang and E.T. Engman, 1980: Aircraft Remote Sensing of Soil Moisture and Hydrologic Parameters, USDA Science and Education Administration, Agricultural Research Results, ARR-NE-8.

Staelin, D.H., 1980: Progress in Passive Microwave Remote Sensing, Nonlinear Retrieval Techniques, Remote Sensing of Atmospheres and Oceans, edited by A. Deepak, Academic Press, 259-274.

Waters, J.W. and D.H. Staelin, 1968: Quarterly Progress Report No. 89, Research Laboratory of Electronics, M.I.T., Cambridge, Mass., 25-28.

Westwater, E.R. and O.N. Strand, 1965: Application of Statistical Estimation Techniques to Ground-Based Passive Probing of the Tropospheric Structure, ESSA Technical Report No. IER 371/1TSA 37.

Wilheit, T.T. and A.T.C. Chang, 1980: An Algorithm for Retrieval of Ocean Surface and Atmospheric Parameters from the Observations of the Scanning Multichannel Microwave Radiometer, Radio Science, 15(3), 525-544.

ACKNOWLEDGMENTS

This study was carried out for NASA/GSFC under Contract No. NAS5-26361. The authors wish to thank Dr. Thomas J. Schmugge, the technical monitor, for providing data of the 1978 Chickasha/Riesel experiment and for his helpful discussions. Additional computer programming and drafting support from S.L. Tung, G.B. Gustafson and A.J. Bussey are also greatly appreciated.

APPENDIX A

FORTRAN CODES OF THE STATISTICAL PARAMETER INVERSION METHOD

```
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C      PROGRAM SPIM
C
C      PURPOSE:
C         STATISTICAL PARAMETER INVERSION METHOD IS USED TO
C         OBTAIN THE INVERSION MATRIX AND PARAMETER ESTIMATIONS
C
C      DISCRIPTIONS OF PARAMETERS:
C         NCASE - NUMBER OF KNOWN CASES (MAX = 500, INPUT)
C         NP - NUMBER OF PARAMETERS (MAX = 10, INPUT)
C         ND - NUMBER OF DATA IN A CASE (MAX = 10, INPUT)
C         DFLAG - .TRUE. FOR PARAMETER ESTIMATION, .FALSE. FOR MATRIX
C                 ESTIMATION ONLY
C         NINP - NUMBER OF CASES FOR PARAMETER ESTIMATION
C         EFLAG - .TRUE. FOR MATRIX EVALUATION, .FALSE. OTHERWISE
C         PARA - ARRAY OF PARAMETER
C         DATA - ARRAY OF DATA
C         CORPD - CORRELATION ARRAY OF PARAMETER AND DATA
C         CORDD - CORRELATION ARRAY OF DATA AND DATA
C         DMT - INVERSION MATRIX
C         DET - DETERMINANT OF MATRIX INVERSION
C         LWOK - TEMPORARY WORKING ARRAY FOR MINV
C         MWOK - TEMPORARY WORKING ARRAY FOR MINV
C         PEVA - INPUT PARAMETER ARRAY FOR EVALUATION
C         DIFF - DIFFERENCE BETWEEN ESTIMATED AND INPUT PARAMETER
C                VALUES
C         SDV - STANDARD DEVIATION OF EVALUATION
C
C      SUBROUTINE AND FUNCTION SUBPROGRAMS:
C         INDATA - ROUTINE TO INPUT NCASES OF DATA AND PARAMETERS
C         COREMT - ROUTINE TO CALCULATE CORRELATION MATRICES
C         MINV - ROUTINE TO CALCULATE MATRIX INVERSION
C         GMPRD - MATRICES MULTIPLICATION
C
C      REMARKS:
C         INPUT NCASE,NP,ND,DFLAG,NINP,EFLAG IN FORMAT(/4I5,L2,I5,L2)
C
C      METHOD
C         STATISCAL PARAMETER INVERSION TECHNIQUE IS USED TO GET
C         INVERSION MATRIX.  RESULTS ARE EVALUATED BY COMPUTING
C         STANDARD DEVIATION.
C
C      DESIGNER: H.K.BURKE AND J-H HO
C
C      PROGRAMMER: J-H HO
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
       LOGICAL DFLAG,EFLAG
       INTEGER NP,ND,NCASE,NINP
       DIMENSION PARA(5000),DAT(5000)
       DIMENSION CORPD(100),CORDD(100),DMT(100)
       DIMENSION LWOK(10),MWOK(10)
       DIMENSION DIFF(10),PEVA(10),SDV(10)
       DATA SDV/10*0./
C
C
       READ(5,5000)
  5000 FORMAT(10X)
       READ(5,5001) NCASE,NP,ND,DFLAG,NINP,EFLAG
  5001 FORMAT(3I5,L2,I5,L2)
       WRITE(6,6001) NCASE,NP,ND,DFLAG,NINP,EFLAG
  6001 FORMAT(1H1,4X,6HINPUT:,/,10X,6HNCASE=,I5,/,10X,3HNP=,I8,
```

```
     +            /,10X,3HND=,I8,/,10X,6HDFLAG=,L5,/,10X,5HNINP=,I6,
     +            /,10X,6HEFLAG=,L5)
      READ(5,5000)
      CALL INDATA(PARA,NP,DAT,ND,NCASE)
      CALL COREMT(PARA,NP,DAT,ND,NCASE,CORPD,CORDD)
      CALL MINV(CORDD,ND,DET,LWOK,MWOK)
      IF(DET.EQ.0) STOP
      CALL GMPRD(CORPD,CORDD,DMT,NP,ND,ND)
      IEND=(ND-1)*NP
      WRITE(6,6000)
 6000 FORMAT(//2X,16HMATRIX ELEMENTS:)
      DO 20 I=1,NP
          IEND=IEND+1
          WRITE(6,6002) (DMT(J),J=I,IEND,NP)
 6002     FORMAT(2X,10E11.4)
   20     CONTINUE
      IF(.NOT.DFLAG) GO TO 100
          READ(5,5000)
          IEND=0
          DO 40 I=1,NINP
             ITEM=IEND+1
             IEND=IEND+ND
             READ(5,5002) (DAT(J),J=ITEM,IEND)
 5002        FORMAT(10F8.3)
   40        CONTINUE
          CALL GMPRD(DMT,DAT,PARA,NP,ND,NINP)
          IF(EFLAG) READ(5,5000)
          IEND=0
          IEND2=0
          DO 60 I=1,NINP
             ITEM=IEND+1
             IEND=IEND+ND
             ITEM2=IEND2+1
             IEND2=IEND2+NP
             WRITE(6,6003) (DAT(J),J=ITEM,IEND)
 6003        FORMAT(//4X,11HINPUT DATA:,/2X,10F11.4)
             WRITE(6,6004) (PARA(J),J=ITEM2,IEND2)
 6004        FORMAT(4X,12HOUTPUT PARA:,/2X,10F11.4)
             IF(.NOT.EFLAG) GO TO 60
                READ(5,5002) (PEVA(J),J=1,NP)
                DO 50 K=1,NP
                   DIFF(K)=PARA(ITEM2+K-1)-PEVA(K)
                   SDV(K)=SDV(K)+DIFF(K)**2
   50              CONTINUE
                WRITE(6,6005) (PEVA(J),J=1,NP)
 6005           FORMAT(4X,12HKNOWN  PARA:,/2X,10F11.4)
                WRITE(6,6006) (DIFF(J),J=1,NP)
 6006           FORMAT(2X,5HDIFF:,46X,10F6.2)
   60           CONTINUE
          IF(.NOT.EFLAG) GO TO 100
              DO 80 K=1,NP
                 SDV(K)=SQRT(SDV(K)/NINP)
   80            CONTINUE
              WRITE(6,6007) (SDV(K),K=1,NP)
 6007         FORMAT(///,22HEVALUATION SDV VALUES:,/2X,10F11.4)
  100     CONTINUE
      STOP
      END
```

```
      SUBROUTINE INDATA(PARA,NP,DAT,ND,NCASE)
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C     SUBROUTINE INDATA
C
C     PURPOSE:
C        INPUT DATA AND PARAMETER FOR INVERSION MATRIX CALCULATION
C
C     USAGE: CALL INDATA(PARA,NP,DAT,ND,NCASE)
C
C     DISCRIPTION OF PARAMETERS:
C        PARA - ARRAY OF PARAMETERS (OUTPUT)
C        DAT- ARRAY OF DATA (OUTPUT)
C        NCASE - NUMBER OF CASES (INPUT)
C        NP - NUMBER OF PARAMETERS FOR A CASE (INPUT)
C        ND - NUMBER OF DATA FOR A CASE (INPUT)
C
C     SUBROUTINE AND FUNCTION SUBPROGRAMS:
C        NONE
C
C     REMARKS:
C        INPUT A CASE AT A TIME, USING FORMAT(10F8.3) FOR DATA FIRST,
C        THEN INPUT PARAMETER BY FORMAT(10F8.3) NEXT LINE
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      DIMENSION PARA(NP,NCASE),DAT(ND,NCASE)
      WRITE(6,6001)
 6001 FORMAT(1H1,5X,12HINPUT CASE :/)
      DO 20 I=1,NCASE
         READ(5,5001) (DAT(J,I),J=1,ND)
         READ(5,5001) (PARA(J,I),J=1,NP)
 5001    FORMAT(10F8.3)
         WRITE(6,6002) (DAT(J,I),J=1,ND)
         WRITE(6,6003) (PARA(J,I),J=1,NP)
 6002    FORMAT(/2X,4HDAT:,/2X,10F11.4)
 6003    FORMAT(2X,5HPARA:,/2X,10F11.4)
   20    CONTINUE
      RETURN
      END
```

```
      SUBROUTINE COREMT(PARA,NP,DAT,ND,NCASE,CORPD,CORDD)
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C     SUBROUTINE COREMT
C
C     PURPOSE
C        TO CALAULATE THE CORRELATION MATRICES FROM 'NCASE' SET OF
C        KNOWN PARAMETERS AND DATA
C
C     USAGE
C        CALL COREMT(PARA,NP,DAT,ND,NCASE,CORPD,CORDD)
C
C     DISCRIPTION OF PARAMETERS:
C        PARA - ARRAY OF PARAMETERS
C        NP - NUMBER OF PARAMETERS FOR A CASE
C        DAT - ARRAY OF DATA
C        ND -NUMBER OF DATA FOR A CASE
C        NCASE - NUMBER OF CASES OF DATA
C        CORPD - CORRELATION ARRAY OF PARAMETER AND DATA
C        CORDD - CORRELATION ARRAY OF DATA AND DATA
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED: NONE
C
C     REMARKS: NONE
C
C     METHOD:
C        CORPD(I,J)=SUM OF PARA(I,K)*DAT(J,K) OVER K (NCASE)
C        CORPD(I,J)=SUM OF DAT(I,K)*DAT(J,K) OVER K (NCASE)
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      DIMENSION PARA(NP,NCASE),DAT(ND,NCASE)
      DIMENSION CORPD(NP,ND),CORDD(ND,ND)
C     INITIALIZATION
      DO 50 J=1,ND
         DO 10 I=1,NP
            CORPD(I,J)=0.
   10       CONTINUE
         DO 20 I=1,ND
            CORDD(I,J)=0.
   20       CONTINUE
   50    CONTINUE
C
      DO 200 K=1,NCASE
         DO 150 J=1,ND
            DO 120 I=1,NP
               CORPD(I,J)=CORPD(I,J)+PARA(I,K)*DAT(J,K)
  120          CONTINUE
            DO 140 I=1,ND
               CORDD(I,J)=CORDD(I,J)+DAT(I,K)*DAT(J,K)
  140          CONTINUE
  150       CONTINUE
  200    CONTINUE
C
      DO 300 J=1,ND
         DO 220 I=1,NP
           CORPD(I,J)=CORPD(I,J)/NCASE
  220      CONTINUE
         DO 240 I=1,ND
           CORDD(I,J)=CORDD(I,J)/NCASE
  240      CONTINUE
  300    CONTINUE
      RETURN
      END
```

```
      SUBROUTINE MINV(A,N,D,L,M)
C ***********************************************************************
C ***********************************************************************
C *
C *          NAME:      MATRIX INVERSE UTILITY (MINV)
C *
C *        PURPOSE:      TO PERFORM MATRIX INVERSION USING THE STANDARD
C *                      GAUSS-JORDAN METHOD. THE DETERMINANT IS ALSO CAL-
C *                      CULATED. A DETERMINANT OF ZERO INDICATES THAT THE
C *                      MATRIX IS SINGULAR.
C *
C *     INTERFACES:   CALLING MODULES SHOULD HAVE A, L, M PROPERLY
C *                   DIMENSIONED.
C *
C *                    CALLED MODULES -  NONE
C *
C *                  INPUT PARAMETERS -  A = INPUT MATRIX
C *                                      N = ORDER OF MATRIX A
C *                 OUTPUT PARAMETERS -  D = RESULTANT DETERMINANT
C *                                      L = WORKING MATRIX
C *                                      M = WORKING MATRIX
C *                                      A = OUTPUT MATRIX
C *
C *                 COMMON BLOCKS SET -  NONE
C *
C *                COMMON BLOCKS READ -  NONE
C *
C *                       DATA FILES -   NONE
C *
C *       COMMENTS:      IBM APPLICATION PROGRAM
C *                      SYSTEM/360 SCIENTIFIC SUBROUTINE PACKAGE
C *                      (360A-CM-03X) VERSION III
C *                      PROGRAMMER'S MANUAL
C *
C ***********************************************************************
C ***********************************************************************
      DIMENSION A(1),L(1),M(1)

C **********        SEARCH FOR LARGEST ELEMENT                      *****

      D = 1.0
      NK = -N
      DO 195 K=1,N
      NK = NK+N
      L(K) = K
      M(K) = K
      KK = NK+K
      BIGA = A(KK)
      DO 115 J=K,N
      IZ = N*(J-1)
      DO 115 I=K,N
      IJ = IZ+I
  105 IF (ABS(BIGA) - ABS(A(IJ))) 110,115,115
  110 BIGA = A(IJ)
      L(K) = I
      M(K) = J
  115 CONTINUE

C **********       INTERCHANGE ROWS                                 *****

      J = L(K)
      IF (J-K) 130,130,120
```

```
  120 KI = K-N
      DO 125 I=1,N
      KI = KI+N
      HOLD = -A(KI)
      JI = KI-K+J
      A(KI) = A(JI)
  125 A(JI) = HOLD

C **********     INTERCHANGE COLUMNS                                  *****

  130 I = M(K)
      IF (I-K) 145,145,135
  135 JP = N*(I-1)
      DO 140 J=1,N
      JK = NK+J
      JI = JP+J
      HOLD = -A(JK)
      A(JK) = A(JI)
  140 A(JI) = HOLD

C **********     DIVIDE COLUMN BY MINUS PIVOT(BIGA)                   *****

  145 IF (ABS(BIGA-.0001).GT.0.) GO TO 155
  150 D = 0.0
      WRITE(6,1000)
 1000 FORMAT(///,5X,16H SINGULAR MATRIX,/5X,12H DETERMINANT,
     +       17H LESS THAN 0.0001)
      RETURN
  155 DO 165 I=1,N
      IF (I-K) 160,165,160
  160 IK = NK+I
      A(IK) = A(IK)/(-BIGA)
  165 CONTINUE

C **********     REDUCE MATRIX                                        *****

      DO 180 I=1,N
      IK = NK+I
      HOLD = A(IK)
      IJ = I-N
      DO 180 J=1,N
      IJ = IJ+N
      IF (I-K) 170,180,170
  170 IF (J-K) 175,180,175
  175 KJ = IJ-I+K
      A(IJ) = HOLD*A(KJ)+A(IJ)
  180 CONTINUE

C **********     DIVIDE ROW BY PIVOT                                  *****

      KJ = K-N
      DO 190 J=1,N
      KJ = KJ+N
      IF (J-K) 185,190,185
  185 A(KJ) = A(KJ)/BIGA
  190 CONTINUE

C **********     PRODUCT OF PIVOTS                                    *****

      D = D*BIGA

C **********     REPLACE PIVOT BY RECIPROCAL                          *****
```

```
      A(KK) = 1.0/BIGA
  195 CONTINUE
C **********      FINAL ROW AND COLUMN INTERCHANGE                    *****
      K = N
  200 K = (K-1)
      IF (K) 100,100,205
  205 I = L(K)
      IF (I-K) 220,220,210
  210 JQ = N*(K-1)
      JR = N*(I-1)
      DO 215 J=1,N
      JK = JQ+J
      HOLD = A(JK)
      JI = JR+J
      A(JK) = -A(JI)
  215 A(JI) = HOLD
  220 J = M(K)
      IF (J-K) 200,200,225
  225 KI = K-N
      DO 230 I=1,N
      KI = KI+N
      HOLD = A(KI)
      JI = KI-K+J
      A(KI) = -A(JI)
  230 A(JI) = HOLD
      GO TO 200
  100 RETURN
      END
```

```
      SUBROUTINE GMPRD(A,B,R,N,M,L)
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C     SUBROUTINE GMPRD
C
C     PURPOSE
C        MULTIPLY TWO MATRICES TO FORM A RESULTANT GENERAL MATRIX
C
C     USAGE
C        CALL GMPRD(A,B,R,N,M,L)
C
C     DESCRIPTION OF PARAMETERS
C        A - NAME OF FIRST INPUT MATRIX
C        B - NAME OF SECOND INPUT MATRIX
C        R - NAME OF OUTPUT MATRIX
C        N - NUMBER OF ROWS IN A
C        M - NUMBER OF COLUMNS IN A AND ROWS IN B
C        N - NUMBER OF COLUMNS IN B
C
C
C     REMARKS
C        ALL MATRICES MUST BE STORED AS GENERAL MATRICES
C        MATRIX R CANNOT BE N THE SAME LOCATION AS MATRIX A
C        MATRIX R CANNOT BE N THE SAME LOCATION AS MATRIX B
C        NUMBER OF COLUMNS OF MATRIX A MUST BE EQUAL TO NUMBER OF
C        ROWS OF MATRIX B
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C        NONE
C
C     METHOD
C        THE M BY L MATRIX B IS PREMULTIPLIED BY THE N BY M MATRIX
C        A AND THE RESULT IS STORED IN THE N BY L MATRIX R.
C
C     SYSTEMS
C        IBM/360 IN SCIENTIFIC SUBROUTINE PACKAGE
C        ADOPTED BY J-H HO.
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C
      DIMENSION A(1),B(1),R(1)
      IR=0
      IK=-M
      DO 10 K=1,L
      IK=IK+M
      DO 10 J=1,N
      IR=IR+1
      JI=J-N
      IB=IK
      R(IR)=0
      DO 10 I=1,M
      JI=JI+N
      IB=IB+1
   10 R(IR)=R(IR)+A(JI)*B(IB)
      RETURN
      END
```

APPENDIX B

RADIOMETRIC RESPONSES AS A FUNCTION OF SURFACE SOIL MOISTURE CONTENT (Chickasha/Riesel, 1978)



Figure B-1 The Ku band (1.67 cm) radiometer data (look angle = 0° and 40°) as a function of volumetric soil moisture content (%) in the 0-2.5 cm layer



Figure B-2 The C band (6 cm) radiometer data (look angle = 0° and 40°) as a function of volumetric soil moisture content (%) in the 0-2.5 cm layer



Figure B-3 The L band (21 cm) radiometer data (look angle = 0° and 40°) as a function of volumetric soil moisture content (%) in the 0-2.5 cm layer



Figure B-4 The PMIS (2.8 cm) radiometer data (look angle = 49.2°) as a function of volumetric soil moisture content (%) in the 0-2.5 cm layer